# Rendering a 3D volume with Image3D

I have 2D JPEG images, of 1024 x 1024 pixels, corresponding to a 500 x 500 square micron section. These images are taken from slices of tissue, each slice seperated by 10 microns. I've been trying to use Image3D to render these images, and have imported all the images, labelling them h1 to h21. I'm having some difficulty telling Mathematica what separation they should have, but I have tried using BoxRatio to get around this. So far I have;

 H = Image3D[{h1, h2, h3, h4, h5, h6, h7, h8, h9, h10, h11, h12, h13,
h14, h15, h16, h17, h18, h19, h20, h21}, BoxRatios -> {1, 1, 0.21}];

Show[H, BoxRatios -> {1, 1, 0.42}]


Which displays out an image like this;

This shows the surface information, but seems to hide any of the internal information. I tried using Opacity[0.5] initially, but this did not work. Then I tried

 ColorFunction->Opacity[0.5]


and this didn't work any better. Is there a way to make the slices opaque, and more importantly, is there a better method to read in and render the images? Thanks!

• – shrx
Jan 23, 2015 at 15:43
• Thanks! Tried that there but ColorFunction context menu is grayed out. Hmmm...
– DRG
Jan 23, 2015 at 15:49
• There's built in ColorFunction editors, I have a post on Wolfram Community that might be helpful for you. See here. Jan 23, 2015 at 16:41
• @Chip : That looks really interesting, but even with options (XRay, LowRange etc) the option is completely grayed out. Could it be due to the JPEG images I'm using have no specified Z components or them being RBG? I can convert them to mono-channel if required..
– DRG
Jan 23, 2015 at 17:04
• @DRG Is it possible to somehow link to your data so we can play with the 3D image? Jan 23, 2015 at 18:06

Image3D is great, but if you want to have fine-tuned control over the opacity then you need to go to DensityPlot3D, or, in this case, ListDensityPlot3D. These are new functions in version 10.2.

In the absence of OP's data, we'll have to invent some data. OP has a set of slices that together make up a 3D object. We can mimic this using Image3DSlices, and we'll use one of the examples from that page.

Get["http://pastebin.com/raw/x1BY2cpu"]
i


The slices are

 Image3DSlices[i]


Now we could just feed these slices back to Image3D to get the original image, which is essentially what OP has done. But what I'd like to do is instead turn this into numeric data,

list = ImageData /@ Image3DSlices[i];


Now we need to deal with the opacity of the 3D density plot. DensityPlot3D and ListDensityPlot3D can take both an OpacityFunction and a ColorFunction, each of which has Scaling option. By default, the plot will feed scaled values to the opacity function, values between 0 and 1, and map these to opacity values that are also between zero and one. I have toyed around and I like using an exponential opacity function with an adjustable parameter. Here is what it looks like with a few different values of this parameter

Plot[Evaluate[(Exp[# f] - 1)/(E^# - 1) & /@ {1, 4, 8, 12}], {f, 0, 1}]


So the blue curve will scale the opacity nearly linearly with the function value (the brightness of the image), while the red curve will make only the largest values opaque.

Now let's use this opacity function with a density plot, and here I've specified the viewpoint since my machine is slow when manually rotating a 3D density plot,

ListDensityPlot3D[list,
OpacityFunction -> Function[f, (Exp[# f] - 1)/(E^# - 1)],
ColorFunction -> Hue,
ViewPoint -> {-1.18, -0.05, 3.18},
ViewVertical -> {-0.96, 0, 0.27},
ImageSize -> 300] & /@ {1, 4, 8, 12}


By adjusting the opacity you can hide or highlight the internal structure. It's far too intensive, at least on my machine, to be able to pair with Manipulate and see the opacity changes in real time, but this may change in future implementations.

Another point, is that since this is a ListPlot variant instead of an Image3D, you can specify the DataRange or change the Ticks in any way you want.